# Simplify out terms that are large

I have something of the form

$$\frac{(A*G+C)(B*G)}{(G^2+junk4)(G + junk3)}$$

I know G is very large, so my largest term will look like:

$$\frac{ABG^3}{G^4} = \frac{AB}{G}$$

And my second largest term will look like $$\frac{CB}{G^2}$$

When expressions get long and complicated, it would be nice to have a function that can perform this calculation for me. Intuitively, I would think this type of simplification using G>>all other terms is pretty useful - so I figured there would be a built in function that can handle this. I haven't found anything so far. Should I try to write my own function to perform this behavior, or is there some built in tools that can do it for me?

• How about the Limit as g-> Infinity? – bill s Feb 4 '19 at 23:55
• Nah, in this case it would just return zero, since the denominator gets larger than the numerator. – Steven Sagona Feb 4 '19 at 23:58

expr = (A*G + C) (B*G)/((G^2 + junk4) (G + junk3));


For large G expand about Infinity

approx1 = Series[expr, {G, Infinity, 2}] // Normal

(* (A B)/G + (B C - A B junk3)/G^2 *)


For small junk3 expand about 0

approx2 = Series[approx1, {junk3, 0, 0}] // Normal // Expand

(* (B C)/G^2 + (A B)/G *)


Or in a single step

approx = Series[expr, {G, Infinity, 2}, {junk3, 0, 0}] // Normal

(* (B C)/G^2 + (A B)/G *)

• Expanding out at infinity is what I'm looking for. Thanks very much. – Steven Sagona Feb 5 '19 at 0:00

The accepted answer is what I'm looking for. But in case anyone ever might be interested, here's a more crude way that I did it if "expressionp" is the name of the expression to simplify.

BiggestTerm[
expression_] := (numerTerms =
CoefficientList[Expand[expression], G];
polynum = Dimensions[numerTerms] - 1;
final = numerTerms[[polynum + 1]]*G^polynum;
First[final])

SimplifyFraction[expression_] :=
BiggestTerm[Numerator[Expressionp]]/
BiggestTerm[Denominator[Expressionp]] // Simplify

SimplifyFraction[Expressionp]